

APPLE II BASIC

语 言 程 序 设 计

(修订本)

张秀越 曹文全 陈寿勤 编著

(下册)

广西区轻工业厅

第十章 图形显示程序设计

APPLE II 具有显示彩色图形的能力。把它的绘图指令编入自编程序中, 可以同时设计程序和绘图, 使显示输出内容更符合使用者的需要。

§ 10.1 图形显示方式

至目前为止, 我们所介绍的各种程序指令和显示状态都是在“文本显示方式”中进行的。所谓文本显示方式, 是指用以显示程序、数据以及操作者与系统会话等方面的工作内容和方式。在这种方式中, 通常进行数值计算和数据处理。至于显示图形, 就必须进入图形显示方式。在图形显示方式中, 可以进行图形和表格的描绘, 这对于各种事务管理、情报检索, 甚至动画游戏等无疑特别有用。

图形显示分低分辨率图形显示和高分辨度图形显示两种。这两种显示的主要区别在于描绘图形的点迹粗细不同。

低分辨率图形显示

APPLE II 在存储器中为低分辨率图形预留两块位置, 每一块称为页。每一个低分辨率的页在荧光屏上有40列乘48行共1920个光点, 每个光点就是一个小的长方形, 图形则用这些小长方形的点构成。而这些点又可以选16种颜色之一种来显示(见COLOR语句)。

低分辨率图形显示可以有两种显示形式: 一种是全屏幕显示, 即一个显示页的48行全部用于显示图形; 可有40列 $\times$ 48行共1920个显示点。另一种是图形文本兼有的显示形式, 在这种显示形式中, 屏幕中最后留出四个文本显示行, 供文本显示使用, 这四行形象地称为“文本窗”。从文本窗中, 可以看到程序, 还可与系统进行对话。由于有了这个文本窗, 所以在低分辨率图形显示情况下, 共有40列乘以40行(即共有1600个图形显示点)供图形显示用。

高分辨度图形显示

高分辨度图形显示也有两页的存储器位置可以利用, 但每一页中能够显示的点的数目却与低分辨率的不同。这种显示方式中, 每一图形页的水平方向分为280个等分, 而垂直方向分为192个等分, 分别是低分辨率图形的7倍和4倍, 所以荧光屏可分为53760个长方形的光点, 即高分辨度的一个点的大小是低分辨率的一个点的二十八分之一。这种细密的点使图形的描绘更为逼真。

在高分辨度图形显示方式, 除了黑白两种之外, 可以使用绿、紫、橙、蓝四种颜色绘图。同样地, 可以将每一显示页设置为全屏幕显示和图形文本兼有的显示方式。后一种显示方式也是在显示页的底部留出四个文本行显示文本。

不论是在低分辨率或是高分辨度图形显示方式, 都可以根据需要将全屏幕图形显示转换

为图形文本兼有的显示方式，或者反之，从图形文本兼有的显示方式转换回到全屏幕图形显示方式。在图形显示程序设计实践中，使用者将会发现，在调试程序时临时把显示方式置成带有文本窗的图形文本显示方式会给调试工作带来好处。

显示方式的设定由有关语句、命令来完成。机器在接通电源、启动运行时，系统规定为文本显示方式，只在执行有关设定图形显示方式的语句、命令之后，才进入图形显示方式。

下面主要叙述图形显示方式中有关的语句、命令和利用这些语句、命令绘制显示图形的一般方法。

§ 10.2 显示方式控制指令

10.2.1 TEXT

格式：

TEXT

该指令的功能是，使系统从图形显示方式转换回到全屏幕文本显示方式，使能显示24行、每行40个字符的文本。此时，光标位置被移到最后一行。

10.2.2 GR

格式：

GR

该指令的功能是，设置低分辨率图形文本显示方式。当这个指令执行后，显示屏上除了低部四行的位置被用来显示文本以外，其余的上部部分设置为低分辨率绘图区，可以显示40行×40列的图形。此区在指令执行后被清除，而光标移到文本窗的第一行。

该指令可以在程序中多次被使用，即使系统是在低分辨率图象显示方式，也可以使用它来清除显示屏。

10.2.3 POKE

为了使显示屏从低分辨率的图形文本显示方式转换为全屏幕图形显示方式，可以输入下面的指令：

POKE -16302, 0

执行了这个指令，文本窗就消失了，显示屏变成了全屏幕图形显示方式。

假如系统目前处于全屏幕图形显示方式，为了转换为低分辨率图形文本显示方式，可以使用上述的GR指令，清除图形荧光幕並重新设置文本窗，也可以使用POKE指令：

POKE -16301, 0

10.2.4 HGR和HGR2

HGR命令的格式是： HGR

该指令清除荧光幕然后使显示转换为高分辨度图形文本显示方式，显示高分辨度屏幕的第一页。屏幕下方留出四个文本行的文本窗，绘图区为280列×160行，相应于存贮区为绘图区第一页。此命令执行后，绘图区被清除为黑色显示屏。此命令的执行不影响HCOLOR命令和文本窗相应的存贮区，可以在这种方式下使用PRINT语句在文本窗内显示文字资料，使得显示屏可以同时显示图形及文字资料。

在这种显示方式，假如需要转换到全屏幕高分辨度图形显示，可以使用命令：

POKE-16302, 0

执行这个命令，就会将文本窗清除，而使绘图区增加32行的图象显示区。

命令HGR2的格式是：

HGR2

它使系统清除荧光屏并转换成高分辨度全屏幕显示方式（280列×192行）且荧幕变成为黑色，而相应的存贮区为绘图区第二页，文本存贮区不受影响。

在执行了HGR2命令之后，可以使用命令：

POKE-16301, 0

使显示方式转换为高分辨度的图象文本显示方式。但要注意，虽然HGR命令也是使显示转换为这种方式，但相应的存贮区则完全不同，HGR所使用的存贮区为第一页，而HGR2所使用的存贮区为第二页。

HGR和HGR2命令执行时，都会把荧幕上的内容清除，这在需要保留原来的高分度辨图象时是十分不方便的。为了进行高分辨度显示方式的转换，同时又不致清除原来的高分辨度图象，可以使用POKE语句来实现。

例如执行下列语句，可以使高分辨度的图形第一页被显示出来而不致将它们先前所存的图形清除掉。

POKE-16304, 0

设定图形方式

POKE-16297, 0

设定高分辨度方式

POKE-16300, 0

选定高分辨度图形第一页（只有从第二页跳回第一页时才需要）

执行下列语句，可以使高分辨度的图形第二页被显示出来而不致将它们先前所存的图形清除掉：

POKE-16304, 0

（当图形方式未被设定时才需要）

POKE-16297, 0

（当高分辨度图形方式尚未被设定时才需要）

POKE-16299, 0

选择高分辨度的第二页

§ 10.3 颜色选择指令及颜色函数

10 3. 1 可显示的颜色及其参数

APPLE II具有彩色图象显示的能力。在低分辨率图象显示的情形，可以使用16种颜色。在高分辨度图象显示的情形，可以使用6种颜色。两种显示方式所能显示的颜色及其对

应的颜色参数用下表表示。

低分辨率图形显示颜色及其参数

颜色参数	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
颜色	黑色	深红	深蓝	紫色	深绿	灰色	蓝色	淡青	棕色	橙色	灰绿	粉红	绿色	黄色	水蓝	白色

高分辨度图形显示颜色及其参数

颜色参数	0	1	2	3	4	5	6	7
颜色	黑色 色1	绿色	紫色 色*	白色 色1	黑色 色2	橙色 色*	蓝色 色*	白色 色2

注：*依电视机而定

10. 3. 2 COLOR语句和HCOLOR语句

COLOR语句和HCOLOR语句分别是选择低分辨率和高分辨度图形显示的颜色。两种语句的格式分别是：

COLOR = <颜色参数>

和 HCOLOR = <颜色参数>

其中，<颜色参数>即§10. 3. 1的表上某种颜色对应的参数数字，它通常为正整数。例如：

COLOR = 13

就将低分辨率图形显示的颜色设定为黄色。

HCOLOR = 6

就将高分辨度图形显示的颜色设定为蓝色。

在高分辨度图形显示的情形，<颜色参数>的范围必须在0~7之间，若超过此范围，则程序执行到这一语句时就会中断，并得出ILLEGAL QUANTITY ERROR的错误信息。

在低分辨率图形显示的情形，颜色参数虽然可以取到255而不致造成错误，但COLOR语句只会取用你选择的数字的低位数。例如：

COLOR = 222

由于222的十六进制数字为DE，它将被认为是COLOR = 14（取十六进制数为0E，E为十六进制数14），结果选择了水蓝色显示图形。

在开机状态，如果没有设定颜色，系统将自动假定选用黑色，即假定颜色号为0。

10. 3. 3 SCRN函数

SCRN函数求荧光屏上某一点位置的颜色值。其格式是：

SCRN (<X坐标表达式>, <Y坐标表达式>)

它将位于括号内座标所标示的点的颜色参数求出来。例如, 程序:

```
10 GR
20 COLOR=10
30 PLOT 12, 24
40 X=SCRN (12, 24)
50 PRINT X
```

执行这个程序段, 就会将屏幕上第13行25列交汇处的点的颜色参数 (由20程序行可知, 颜色参数为10) 放入变量X中, 然后在荧光屏上输出10这个数字。

这个函数在设计较高级的低分辨率图形的程序时会有很大帮助。

注意, <X坐标>, <Y坐标>的值的允许范围均为0~47。若<X坐标>的值在0~39之间, SCRN函数得到的值就是(X坐标, Y坐标)那儿的点的颜色参数; 若X坐标在40~47之间, 而Y坐标在0~31之间时 SCRN函数就得到(X坐标-40, Y坐标+16)那个点的颜色参数; 如果X坐标在40~47之间, 而Y坐标在32~47之间时, SCRN函数得到的值就是一个与荧幕的内容无关的数。

假如目前的显示屏正显示的是文字, 或者是有文本窗, 并且所指的位置正好在文字幕内, 则SCRN函数得到的是这个字符某一个半部的颜色参数; 若Y坐标为偶数, 则对应那位置上字符相应代码后四个二进位值; 若Y坐标为奇数, 则对应那位置上字符相应代码前四个二进位值。在(X, Y)那个文字的ASCII码n可以用公式:

$$n = \text{SCRN}(X, 2 * Y + 1) + 16 * \text{SCRN}(X, 2 * Y)$$

计算, 于是再用CHR\$(n)就能够求出这个文字了。式中X要在0~39, Y要在0~23的范围内。

SCRN函数用于高分辨度图形的颜色参数值计算时, 所得到的数是按低分辨率图形区域计算, 而不是按高分辨度图形区域来算的。

§ 10.4 低分辨率图形设计

在低分辨率图形显示方式中, 可以使用下述绘图指令, 实现在显示屏上画一个单独的点, 画出不同长度的垂直的或水平的直线。这些指令只能使用在低分辨率图形的第一页位置。

10.4.1 PLOT语句

PLOT语句的格式是:

PLOT <X坐标表达式>, <Y坐标表达式>

其中, <X坐标表达式>为算术表达式, 其值在0~39之间; <Y坐标表达式>为算术表达式, 其值在0~47之间。

该语句的功能是, 在指定的座标(X, Y)位置画出一个点 (实际上是一个小长方形), 点的颜色使用这个语句执行前最近一个COLOR指令选定的颜色。

注意，在带有文本窗的图形文本显示方式中，若Y座标表达式的值在40~47的范围，所要画出的点将以字符的形式出现在文本窗中，要恢复该点的本来面貌，可使用POKE-16302, 0命令使显示方式转换为全屏幕图形显示，使全屏幕都成为制图区。

〔例1〕在荧幕上画一条对角线的程序：

```
10 REM DRAW A DIAGONAL LINE
15 REM ACROSS THE LOW-RES SCREEN
20 GR                                     (进入低分辨率图形文本显示方式)
30 COLOR=RND(1)*16                       (产生0~16中任一颜色参数)
40 FOR Y=0 TO 39                         (从点(0,0)开始直至点(39,39)在屏幕上画
50 PLOT Y, Y                             出40个点，组成荧幕上一条对角线。
60 NEXT Y
70 GOTO 30                               (重复画对角线操作，但对角线的颜色不断变换)
```

〔例2〕在全屏幕低分辨率图形显示方式绘制位于(30, 25)座标位置的一个橙色的点。

```
10 GR:POKE-16302,0
20 COLOR=9
30 PLOT 30,25
40 END
```

10.4.2 HLIN语句

HLIN语句为在低分辨率图形显示方式下，在荧幕上画出一条水平直线。该语句的格式是：

HLIN<X座标1>,<X座标2>AT<Y座标>

其中，<X座标1>、<X座标2>和<Y座标>均为算术表达式，它们的取值范围是：X座标为0~39，Y座标为0~47。

该语句的功能是：在<Y座标>的行上，自点(X1, Y)至点(X2, Y)画一条水平直线。所用的颜色由该语句执行前最近的一个COLOR命令选择的颜色所决定。

假如荧幕有文本窗，而且Y座标值在40~47之间，那末HLIN语句就在文本窗中描出一条由文字组成的直线。X座标和Y座标的值超出了各自允许的取值范围就会产生出错信息。

〔例3〕

```
10 GR                                     (进入低分辨率图形文本显示方式)
20 COLOR=3                               (选择使用紫色)
30 HLIN 0,39 AT 0                       (分别在荧幕上绘图区的顶端和底端自最左边到最
40 HLIN 0,39 AT 39                     右边画出一条水平直线，使用紫色)
50 END
```

〔例4〕

```
10 GR:POKE-16302,0
20 FOR Y=38 TO 42
```

```

30 COLOR = Y - 37
40 HLIN 10, 20 AT Y
50 NEXT Y
60 END

```

这个程序中，第10行的作用是进入低分辨率绘图方式，使用全屏幕显示图形，20至50行是循环，在38至42的5个显示行上，自点(10, Y)到点(20, Y)各自画一条水平直线，由这5条相邻的彩色线构成一个长方形的显示块，所用的颜色（色号分别为1, 2, ..., 5）分别为深红、深蓝、紫色、深绿和灰色。

10.4.3 VLIN语句

在低分辨率图形显示方式下，VLIN语句能在荧幕上画一条竖直线。该语句的格式是：

VLIN <Y坐标1>, <Y坐标2> AT <X坐标>

其中<Y坐标1>, <Y坐标2>和<X坐标>均为算术表达式，它们的取值范围是：X坐标为0~39, Y坐标为0~47。

该语句的功能是：在<X坐标>的列上，自点(X, Y1)至点(X, Y2)画一条竖直线，所用的颜色由该语句执行前最近的一个COLOR命令选择的颜色所决定。

假如荧幕有文本窗，而且Y坐标在40~47之间，那末，所画的竖直线段的一部分或全部会以文字出现，而不再是点了。X坐标和Y坐标的值超出了各自允许的取值范围就会产生出错信息。

[例5]

```

10 GR
20 COLOR = 5
30 VLIN 0, 39 AT 0
40 END

```

这个程序会使用颜色代号为5的颜色（灰色）在荧幕的最左边自绘图区的顶部画一条竖直线至绘图区底部。

[例6]

```

10 GR : COLOR = 13
20 VLIN 10, 20 AT 5
30 HLIN 5, 15 AT 10
40 VLIN 10, 20 AT 15
50 HLIN 5, 15 AT 20
60 END

```

这个程序会在低分辨率图形显示方式的绘图区用黄色画一个方框，方框的四个顶点坐标为(5, 10)、(15, 10)、(15, 20)、(5, 20)。

[例7]

```

10 GR : COLOR = 2

```



```

20 VLIN 0, 39 AT 10
30 HLIN 0, 39 AT 25
40 COLOR= 4 : Y=20
50 FOR X=12 TO 30
60 VLIN Y, 20 AT X
70 Y=Y-1
80 NEXT X
90 END

```

这个程序执行时，荧幕上先用深蓝色画一对坐标轴（原点在点(10, 25)），接着用深绿色在坐标系的第一象限（右上部）沿 $Y=20$ 的水平线画出19条垂直线，构成一个深绿色的直角三角形。由于是在低分辨率图象显示方式，该直角三角形的斜边看来有点象锯齿形的。

§ 10.5 高分辨度图形设计 · H PLOT语句

本章的前三节介绍了高分辨度绘图方式的建立、绘图区和绘图页的安排、转换，以及高分辨度图形显示颜色的选择等基本命令，本节介绍高分辨度方式下绘制图形的有关指令。

H PLOT语句

在APPLESOFT高分辨度图形方式，具有在显示屏幕上绘制任何垂直的、平行的或者任何角度的线的功能，就象在绘图纸上描点画线一般地方便。这一功能是由H PLOT指令实现的。

H PLOT指令的格式是：

格式 1：

H PLOT <点座标>

格式 2：

H PLOT TO <点座标>

格式 3：

H PLOT <点1座标> TO <点2座标> [TO <点i座标>]

其中<点座标>、<点1座标>、<点2座标>及<点i座标>的形式是：

<X座标表达式>，<Y座标表达式>

X、Y座标表达式均为算术表达式，其值分别代表某点在荧幕上的行、列号。

这些指令的功能是：

格式 1的指令为在荧幕的坐标点(x,y)上显示一个有色的点，这个点的颜色由上一次执行的H COLOR语句决定；

格式 2的指令为从最近所绘制的点(或线的最末点)开始向<点座标>所代表的点画一条有颜色的直线，所用的颜色由上一次执行的H COLOR语句决定。

格式 3的TO <点i座标>项可以没有，也可以出现一个或多个。该指令的意义是，从<点1座标>开始向<点2座标>画一条直线，如果还有<点3座标>，<点4座标>，…

…，〈点：座标〉，则从所画出的直线末点继续向下一点画线，直至第 i 点止。显然，执行这个指令的结果，通常是在荧幕上画出一条有颜色的折线。所用的颜色由上一次执行的HCOLOR语句决定。

格式3的画线段数没有硬性规定，但所有的点的座标必须在一语句行上写完。

〔例1〕

H PLOT 8, 16

这个语句在目前所选择的高分辨率显示页的第9行第17列交点处使用刚被选择的颜色画出一个点。

〔例2〕

H PLOT 0, 0 TO 279, 191

这个语句从目前所选择的高分辨率全屏幕显示屏上画出一条对角线。所用颜色由这个语句之前刚选定的颜色决定。

使用这种形式的绘图语句，就可以从荧光屏上的任何一点画一条线段到另一点。

〔例3〕

H PLOT 50, 50 TO 230, 50 TO 230, 140 TO 50, 140 TO 50, 50

这个语句在高分辨度绘图区中使用前面刚被选定的颜色绘制出一个长方形。图10—1标出了长方形的顶点座标及画线方向。




图10—1 长方形绘制示意图

〔例4〕绘制一个五边形ABCDE，各点的座标分别是，A：(150, 0)，B：(250, 50)，C(200, 150)，D：(100, 150)，E (50, 50)，试写出绘制这个高分辨度图形的程序，要求使用蓝色画线。

程序如下：

```
10 HGR2: HCOLOR = 2
20 H PLOT 150, 0 TO 250, 50 TO
   200, 150 TO 100, 150 TO 50, 50
   TO 150, 0
30 END
```




图10—2 五边形绘图显示

执行这个程序，就会在高分辨度全屏幕绘图区的第二页上显示出形如图10—2的蓝色图形。

一个绘制图形的要求给出之后，要编制显示该图形的程序，最好先在坐标纸上人工地画出所需的图形或草图，定出各端点坐标，就可利用H PLOT指令的格式3方便地编写出语句。

进入高分辨度图形显示方式时如果使用的是HGR指令，使屏幕设置为高分辨度图形文本显示方式，或者在执行了HGR2命令之后，用了POKE-16, 1, 0命令使显示方式转换为高分辨度图形文本显示方式，由于图象区的有效行数为160行（0~159），因而，当绘图的点坐标中y坐标进入160至191范围内时，则这部分图形就看不见。

〔例5〕在高分辨度屏幕上绘制一个绿色五角星。

方法是，先在坐标纸上按所希望的大小画出五角星草图，确定各顶点坐标，如右图所示。

下面就是一个显示五角星的程序，在高分辨度图形文本显示屏幕上作图。

```
10 REM DRAW A FIVE STAR GRAPH
```

```
20 HOME : HGR
```

```
30 HCOLOR = 1
```

```
40 H PLOT 80, 59 TO 220, 59 TO
```

```
100, 159 TO 150, 9 TO 200,
```

```
159 TO 80, 59
```

〔例6〕在高分辨度图形显示方式屏幕上，以（150，100）为圆心，半径（设相当于75列的距离）参数为75，绘制一个坐标系及圆，然后画一圆形面，如图所示。

圆的图形设计，可以应用参数方程很方便地解决。圆的参数方程是：

$$x = R \cos t$$

$$y = R \sin t$$

可以看出，根据这个参数方程，角度从 $0^\circ \sim 360^\circ$ 旋转一周，便可以得到圆的轨迹。根据要求，在（150，100）处设立坐标原点（即屏幕上第100行150列的交点），我们用H PLOT语句在（150，100）处画两条互相垂直的直线作为坐标轴，然后利用参数方程按角度增量为 2° 算出（x，y）坐标，采用折线逼近方式画出圆，再使用连结圆心和各圆周上的点的方法画出圆图。

程序为：

```
10 HGR2
```

```
20 HCOLOR = 1 : R = 75
```

（进入高分辨度全屏幕图形显示方式）

（选用绿色，半径参数确定为75）




图10—3 五角星绘图显示




图10—4 圆图显示

30 HPLOT 20, 100 TO 250, 100	(画x轴)
40 HPLOT 150, 10 TO 150, 190	(画y轴)
50 X0=150 : Y0=100 : D= 2 * 3.14159/180	(定原点和角度增量)
60 P=X0+R : Q=Y0	(确定画圆起点)
70 FOR T= 0 TO 6.2832 STEP D	
80 X=X0+R * COS (T)	(计算圆周上新点的座标)
90 Y=Y0-R * SIN(T)	
100 HPLOT P, Q TO X, Y	(两点间画一线段)
110 P=X : Q=Y	(设定新的起点)
120 NEXT T	
130 HOME	(清除显示屏)
140 P=X0+R : Q=Y0	
150 FOR T=0 TO 6.2832 STEP D	
160 X=X0+R * COS (T)	
170 Y=Y0-R * SIN(T)	
180 HPLOT X0, Y0 TO X, Y	(从座标原点至圆周上的点画直线段)
190 NEXT T	
200 TEXT	
210 END	

读者如果在机上运行这个程序，就会发现，由于我们采用折线逼近曲线，画出的圆是呈锯齿形的。读者不难将程序改造成用密集的圆点逼近圆周的绘图程序。此外，由于绘图区上的各个点实际上呈小矩形状态（X方向较长，Y方向较短），因而画出的圆呈椭圆状。读者不妨试试将90行和170行计算Y座标的表达式中的R改为（R+10），再运行这个程序，就会发现画出的图形会更为接近圆。

关于高分辨度制图，APPLE机还提供了使用所谓“图形定义”的方法，利用 DRAW、XDRAW、ROT、SCALE、SHLOAD和BLOAD等命令进行造型表的存取和绘图控制。因为较繁琐，这里就不介绍了。

第十一章 APPLE 中文系统

APPLE II 微型电脑配上中文信息处理系统, 就可以直接输入、输出汉字, 使许多原来需要用英文处理或表示的任务、文字可以用汉字代替。这对于在英文尚不普及的中国推广使用电子计算机, 无疑具有重要的意义。

APPLE II 微型电脑中文信息处理系统的核心部件是一块中文字形产生器 (简称汉字卡), 本章介绍汉字卡的功能、使用方法, 汉字的组合方法及输入输出控制等。

§ 11.1 汉字卡简介

APPLE II 机的汉卡有二种: C-PLUS II 汉卡及 C-PLUS II - A 汉卡, 是一块小巧玲珑的中文字形产生器, 它作为一个外部设备插件插在主机插座上, 通过一个很小的缓冲区与主机交换信息。中文系统采用仓颉中文输入法, 以仓颉码24个字母为字库, 通过执行组字程序, 组出所需的汉字。汉字采用 16×16 点阵显示。全屏幕能显示汉字的能力是17字 $\times$ 10行。

C-PLUS II 汉卡包括常用繁体汉字3300个, 有效组合字数约有15000个。

C-PLUS II - A 汉卡包括常用繁体汉字4800个, 有效组合字数约 23000 个, 可直横列印大小字, 字与字间、行与行间的距离, 均可控制选择。

§ 11.2 汉字卡的使用

APPLE II 微型电脑使用C-PLUS II 汉卡时, 必须有 48K 字节的RAM存贮容量和稳定的电源输出装置, 否则不能使用本汉字卡。

使用C-PLUS II 汉卡时, 荧光屏上可显示出17字 $\times$ 10行的中文字 (一个中文字等于2个英文字母的距离, 亦即可显示出34字 $\times$ 10行的英文字母)。

一、外围设备

使用C-PLUS II 汉卡时, APPLE II 机的基本外围设备, 无需作任何改动, 装配容易, 现分述如下:

(1) 打印机部分: EPSON公司出品的MX-80 II 型或 III 型及MX-82, MX-100等型号的打印机均可使用。要打印中文字时, 只需将打印机后方4字位的 DIP 开关中, N0.3 (AUTO FEED) 开关设置于<ON>位置即可(注: 若使用的是C-PLUS II - A 汉卡, 不用调节这个DIP开关)

(2) 磁盘驱动器无需改动。

(3) 键盘部分：不需要使用特殊键盘，亦不需要作任何改动，但为了操作方便，可在键盘上贴中文字母贴纸。

二、汉字卡的安装和开启

安装汉字卡时，一般依照下列步骤，顺序操作：

(1) 首先检查主机电源是否稳定，注意不要在电压波动较大的条件下使用汉卡，以避免损坏。

(2) 将主机电源关掉，并将主机上方外壳取下，这时可看到机器后方有8个扩充接口插座（SLOT），由左至右分别编号为0，1，2，3，4，5，6，7号插座。汉字卡可插在1~7号中的任何一个扩充插座，最好置于3号扩充插座上。第0号插座有特殊功用，不可插汉字卡。插汉卡时应将汉卡垂直插进。注意不可在电源接通状态下操作。

(3) 检查汉卡安装是否正确。如安插工作一切正常，就可将主机电源打开。电源开启后，机器的起动情况和屏幕显示内容应与未装汉卡时的情形相同。若有不同之处，应立即关闭整个系统的电源，并且重复安装操作过程，直到各种情形均正常为止。

(4) 打开电源后，如果系统起动正常，就可直接敲入PR#CN指令，并执行之。这里CN表示汉卡所插的插座编号。若如上所述，将汉卡插在3号插座上，则输入PR#3(↵)，开始执行PR#CN动作后，在荧光屏上显示如下字幕：

*** 佳佳C-PLUS II 汉卡 ***

中文字形产生器专利公告号码43133

1982/07/20 改良代号××

序号：××××

此即表示系统已进入中文状态，便可进行中文信息处理。〔注〕

(5) 至目前为止，操作情况一切正确，现在可将机器的上盖板盖上。注意盖外壳时，最好先关闭电源，以免不意碰到任何组件，造成不必要的损坏。

(6) 从现在开始，你就可以输入程序，操作机器，让汉卡工作。汉字卡的中文输入，是利用键盘上英文字母A—Y键，代表中文字母来组字的，由CTRL—L（CTRL键和L键同时按下）来决定是英文字母或中文字母。

当按下CTRL—L时，荧光屏的左下角显示出“中文”两个小字，同时发出“哔”声，提示你现在开始接受中文输入，这时A—Y键代表中文字母“日一卜”。

中文输入完毕，再按CTRL—L时，亦发出“哔”声，同时左下角“中文”二字消失，键盘恢复为英文输入状态。通过这种方法，就可以交替地输入中英文字。

(7) 当荧光屏左下角显示“中文”二个小子，表示系统处于中文输入方式之后，每输入一中文字母，均会在“中文”的右边依次显示。每个中文字取码最多取五码（五个中文字母），输入完毕，再按空格键，假如汉字的拼写有效，就会组出一个汉字显示出来，而原先

〔注〕该字幕的汉、产、专、号、码各字实际显示为中文繁体字。

的中文字母消失,等待下一字的输入。若输入错误的中文字码,或每个字超过5个字母,均会发出“哔”声,警告使用者输入的字母无效。

(8) CTRL—L的功能操作熟练后,可以利用CTRL—D之功能来选择左下角的二个小字“中文”显示与否,CTRL—D是以开关方式使用的,可交替互换。

输入中文字之前,先按CTRL—D两键,再按CTRL—L两键,这时荧光屏上没有任何改变,左下角不显示“中文”二个小字。但其接受的键是中文字母。如要回复到英文输入方式,可先按CTRL—L再按CTRL—D,就可回复原始状态。

§ 11.3 中文状态的软件功能

C—PLUS汉字卡适用于在APPLESOFT BASIC语言环境中工作,但在C—PLUS汉字卡执行中,有下列指令不能使用:

TEXT, VTAB, HTAB, HOME, TAB, POS, SPC, FLASH,
INVERSE, NORMAL, GR, COLOR, PLOT, HLIN, VLIN,
SCRN, HGR.

除了上述17个指令之外,APPLESOFT中的所有BASIC指令,均保持其原有性能,可自由选用。

然而,在上述指令中,除了用于绘图、游戏控制的指令外,在中文状态下,大部分指令(例如VTAB, HTAB, HOME等指令)可用POKE命令干预原内存地址的状态和用CALL调用命令把某子程序调出来帮助完成在中文状态下不能使用的指令原有的功能。POKE和CALL可作程序语句,也可作立即执行的命令用。

11.3.1 CALL 调用语句的应用

在中文状态下,可提供五个CALL命令,由用户任意选择使用。为便于记忆,可先将调用地址进行定义,要使用时,可直接将后面算式结果代入CALL的地址名即可。

(1) CALL HM (HM定义地址名,下类同);此命令可取代文本方式语言中HOME的功能,当执行该命令后,即可清除屏幕,并将光标横线移至左上角。

(2) CALL CC; (CLEAR CURSOR) 此命令执行时,会清除先前PRINT指令所遗留的光标横线,以美化显示屏。

(3) CALL CE; (CLEAR TO END OF LINE) 此命令功能与文本方式的指令<ESC E>相同,可清除光标横线开始的整行字幕。

(4) CALL CL; (CLEAR LINE) 此命令可清除光标横线所在的该行显示。

(5) CALL CF; (CLEAR TO END OF SCREEN) 此命令与文本方式指令<ESC F>的功能相同,是清除光标横线之后的整个屏幕显示。

上列各个命令的地址与C—PLUS II 汉字卡所插入的主机插座号码有关,若用CN表示汉字卡所插入的插座号码值,则各命令式的地址计算公式如下所示:

$$BS = 49152 + 256 * CN \quad (\text{五个命令式的地址值均以此为基数})$$

$HM = BS + 21$

$CC = BS + 24$

$CE = BS + 36$

$CL = BS + 39$

$CF = BS + 42$

假如汉字卡插在第3号插座上, 现在想实现HOME的功能, 那末首先要定义HOME命令的地址, 亦即输入 $HM = 49152 + 256 * 3 + 21$, 执行时使用命令CALL HM即可。又如, 想要清除PRINT指令遗留的光标横线, 必须先输入 $CC = 49152 + 256 * 3 + 24$, 然后使用CALL CC即可使该横线消失。

11.3.2 POKE存贮地址的使用

在中文状态下, 提供多个POKE存贮地址语句, 供用户根据需要选择使用。这些存贮地址与打印机、汉字卡所在的插座号值有关, 现以CN代表汉字卡所在插座值, PT代表打印机所在插座值。下面分类说明:

- (1) POKE 1400 + CN, PT, 表示启动打印机, 准备打印。此命令与APPLESOFT BASIC语言中PR#PT的功能相同。

POKE 1400 + CN, 0: 表示关闭打印机, 禁止打印。

- (2) POKE 1656 + CN, 0: 表示打印机印出大字横印。

例: 佳佳漢卡佳佳漢卡

POKE 1656 + CN, 1: 表示打印机印出大字直印

例: 世世漢卡世世漢卡

POKE 1656 + CN, 2: 表示打印机印出小字横印

例: 佳佳漢卡佳佳漢卡

POKE 1656 + CN, 3: 表示打印机印出小字直印

例: 世世漢卡世世漢卡

注: 如果不加选择, 则设定为大字横印。

- (3) POKE 1912 + CN, N: 表示在打印机工作时, 定义各个ASCII字符之间的距离为N个点的间隔, 中文字与中文字之间的距离为2N个点。例如, 若 $N = 2$, 则英文与英文文字之间的距离为2个点, 汉字与汉字之间的距离为4点。如果不定义N值, 则N设定为0。

- (4) POKE 1784 + CN, N: 表示在打印机工作时, 定义印出的行与行之间的距离为N个点的间隔, 如果不定义N值, 则自动设定 $N = 4$ 。

- (5) POKE 2040 + CN, N: 表示定义打印机每行允许打印的字数, 以英文字为单位, 中文字占2个英文字的位置, 例如, $N = 80$, 则每一行可

印80个英文字，或是40个中文字。不同型号的打印机所能印出的行字数不同，MX—80可使用至120个英文字数，MX—100则可达到240个左右，如果不定义N值，则自动设定N=50。

(6) POKE 214, H: 与文本方式的APPLESOFT BASIC语言中的HTAB的功能相同，H值为移动光标横线位置的水平座标值，最左方为0，最右方为33。

(7) POKE 215, V: 与文本方式的APPLESOFT BASIC中的VTAB的功能相同，V值为移动光标横线位置的垂直座标值，最上方为0，最下方为9。

(8) POKE 253, 0: 此功能与CTRL—D相同，在显示中文字时，左下角不会产生小的“中文”及字母。

(9) POKE 253, 255: 恢复原始状态，即在显示中文时，左下角能产生“中文”及字母小字。

注：在使用C—PLUS II—A汉字卡时，以上所列各项功能均具备，在使用C—PLUS I汉字卡时，第二、三、四、五项功能不具备。

例：如果打印机控制卡插在1号插座，汉字卡插在3号插座，且用小字横印出程序，其步骤如下：

1.] PR#3 (按RETURN键)
2.] POKE 1403, 1 (按RETURN键，开启打印机)
3.] POKE 1659, 2 (按RETURN键，此行已打印在打印纸上)
4.] LIST (按RETURN键，开始打印程序，直到打印结束)

不再使用打印机时，必须发POKE 1403, 0命令，关闭打印机。

例：下面是一个中文自动启动程序，可加在程序最前面，这样就不必每次打PR#CN，亦可当作HELLO程序处理。

程 序：

```
5 REM: CN 表示汉字卡所在的插座号码值
10 HIMEN: 36864
20 POKE 37984, 0
30 PR#CN
40 PRINT ""
50 POKE 43603, 3
60 POKE 43604, 192+CN
70 POKE 43605, 48
80 POKE 43606, 192+CN
90 POKE 54, 189
100 POKE 55, 158
```